
FAX SENDING EXAMPLE

Note: This technique only works on Mac OS X 10.3, and assumes you have a telephone line
connected into the Internal Modem (which is not being used to log in at the same time...)

Kevin Cunningham, kcunning@alum.mit.edu, 11/11/2004

recordID 104

fax_recipient_name Dr. Rupert Boardwalk

fax_number (617) 499-2345

fax_message By the way, Rupert, it was great seeing
you at the Flagon Fair last Thursday. It’s
always great to run across old friends
like that.

Send Fax

fax_sent_timestamp

11/11/2004 12:36:49

fax_print_file_path /Users/kcunning/Desktop/

fax_print_file_name temp.pdf

Make sure to re-Specify
the options of the Print
command in the
Send_Fax script to match
the values you enter
here...



FAX MESSAGE

To: Dr. Rupert Boardwalk
Fax: (617) 499-2345
From: Kevin Cunningham
Subject: About Your Account
----------------------------

Hello.

This is just a brief reminder that your account payments are due in full
by the last day of each month.

By the way, Rupert, it was great seeing you at the Flagon Fair last
Thursday. It’s always great to run across old friends like that.

We enjoy doing business with you.

Thanks for your attention to this matter.

See you next month,
--Kevin



Send_Fax

#Send_Fax goes to a layout that will be be used to generate a PDF file, which will then be sent as a fax to the
recipient identified in current record

#Test that required fields are not empty

If [ IsEmpty(FAX::recordID) or
IsEmpty(FAX::fax_number) ]

Show Custom Dialog [ Title: "WARNING"; Message: "You need to include a fax number. Canceling."; Buttons: "OK" ]

End If

Freeze Window

#Go to target layout

Go to Layout [ "FAX_fax_screen" ]

#Set printout to assume standard paper size

Print Setup [ Orientation: Portrait; Paper size: 8.5" x 11" ]
[ Restore; No dialog ]

#Use print driver to generate PDF. Specify "Save as PDF File..." in Print window rather than printing to hardcopy.
Also, use the name

#Also, make sure that the file path/name set here is also set in main FAX table.

#If you need to specify a new path, click the Specify button for the Print step and select Save as PDF...

Print [ All Pages; Orientation: Portrait; Paper size: 8.5" x 11" ]
[ Restore; No dialog ]

#Take the printed PDF file, submit it to the fax queue, then delete the file

Perform AppleScript [ Calculated AppleScript: Let(
  [

  /* filename of PDF file already generated */
  path = Case(Right(FAX::fax_print_file_path; 1) = "/"; ""; "/") & FAX::fax_print_file_path;
  printfilefullname = path & FAX::fax_print_file_name;

  /* new name to give print file so it will be distinguishable in FAX queue */
  /* here, will be of the form "2004_11_10_rec123_Dr_Svengali.pdf"  */
  /* but can be any effectively unique string */
  /* (probably should include some form of time in name to distinguish two faxes from the same record) */
  rightnow = Get(CurrentDate);
  date_stamp = Year(rightnow) & "_" & Right("0" & Month(rightnow); 2) & "_" & Right("0" & Day(rightnow); 2);
  recordID = FAX::recordID;
  nameref = Substitute(Filter(FAX::fax_recipient_name; " ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz"); "
"; "_");
  faxfilename = path & date_stamp & "_rec" & recordID & "_" & nameref & ".pdf";

  /* fax number to use */
  fax_number_orig = FAX::fax_number;
  faxnum = Filter(fax_number_orig; "0123456789")   /* remove non-numeric characters from fax phone string */

  ]
;

  /* execute three-step shell script:
        1) rename (mv) file to unique name
        2) send file to fax queue
        3) remove (rm) file
  */
  "do shell script \"mv " & printfilefullname & " " & faxfilename & " ; lp -d Internal_Modem -o phone=" & faxnum & " " & faxfilename &
" ; /bin/rm " & faxfilename & " \" "

) ]

#Mark time of fax, to record the fact that a fax was sent

November 11, 2004 0:37:34 -1-fax.fp7 - Send_Fax



Send_Fax

#Mark time of fax, to record the fact that a fax was sent

Set Field [ FAX::fax_sent_timestamp; Get(CurrentTimeStamp) ]

Commit Records/Requests

Go to Layout [ original layout ]

